
概要 
本研究では属性への確率的な注目と代替案の反復的

消去を用いて意思決定者の認知プロセスをモデル化し，
実際にコンピュータプログラムによるシミュレーショ
ン実験を行った．この結果，文脈効果やギャンブル比
較のアノマリー例題の多くを基本的に一つのモデルで
実験的に再現できる可能性があることが示された． 
 
キーワード：確率的な属性への注目, 反復消去，文脈
効果，意思決定と推論のアノマリー 

1. はじめに 

選択肢がいくつかあって，その中から一つを選ぶよ

うな課題に対して，人々はどのような認知プロセスで

候補を絞り込んだり，評価を下したりするのだろうか．

意思決定者の認知プロセスの研究は，たんに選択結果

を予測するだけではなく，意思決定者が環境のどの側

面を注目したり注目しなかったりするか，またどのよ

うにその最終判断に影響するのかを予測できる理論を

必要とする．このような選択問題における属性への確

率的な注目とそれによって代替案を絞り込む認知プロ

セスのことを，本論文では情報反応と呼ぶ． 

より一般的な文脈においては，情報反応とは生体や

ソフトウェアロボットが環境の変化をシグナルとして

察知し，実行できる行動を選んで適用できるかどうか

素早く判断する状況適合を意味する．本研究では選択

問題における情報反応を，次のような手続きの反復と

してモデル化する．すなわち各時点である属性がラン

ダムに注目され，その属性についての評価が許容され

ない，すなわちその属性にかんしてその時点で残って

いる他の代替案に対して何らメリットがないものを消

去する．  

また異なるジャンルのよく知られた例題に対して提

案する手続きを適用する．条件文推論，消費者ブラン

ド選択における文脈効果，ギャンブル選択（リスク選

択），時間選択の４領域にわたって文献で報告されたア

ノマリー現象が再現されるか実験的に確かめる． 

図 1  確率的な反復消去アルゴリズム 

 

本研究では，属性への確率的な注目に基づき選択す

る候補を絞り込むヒューリスティックな手続きを提案

し，異なる複数の領域のサンプル例題に対して適用す

る．以下ではこれを反復消去アルゴリズムあるいは，

たんに反復消去と呼ぶことにする．図 1 にその疑似コ

ードを示す．サンプル例題は条件文推論[17]，消費者ブ

ランド選択[2,6,9,10,12,14]，ギャンブル比較[1,3,4,7,8,15]，

時間選択[11]という行動選択の４領域にわたる． 

反復消去アルゴリズムへの入力は，与えられた選択

問題の問題文章や図表から属性とその価値の組に変換

された問題表現のデータである．提案する手法は多属

性意思決定問題として述べられていない問題に対して

も適用可能である．これについては，第 2 節で条件文

推論の課題であるカード選択課題をモデル化する例を

用いて説明したい．ここでは図 1 に示された手順を簡

潔に説明しておこう．まず「候補リスト」にすべての

属性がセットされ，そこから類似性を考慮して，ラン

ダムに注目される属性が選ばれる．選択者が注目した

属性は「注目履歴」に登録される．つぎに注目した属

性と候補リストから，他の候補に比べて魅力のないも

のとして，非許容的な代替案を見つけて削除リストに

追加する．候補リストから削除リストに挙がった候補

を削除し，注目履歴に注目した属性とそれに基づいて

Function 反復消去( 代替案リスト, 属性リスト ) 

候補リスト=代替案リスト; 

注目履歴=[ ]; 

While 候補リスト=[ ] is false 

注目属性=類似性に基づく確率選択(属性リスト) 

削除リスト=非許容(注目属性, 候補リスト) 

候補リスト=remove(候補リスト, 削除リスト) 

注目履歴=append(注目履歴, 注目属性:削除リスト) 

If 先頭要素(注目履歴).length =1  

Return 先頭要素(注目履歴) 

 Else Return ランダム選択(先頭要素(注目履歴)) 

削除された候補を記録する．これをを候補リストが空

になるまで繰り返す． 

上で述べた手続きは，要するに非許容な代替案を見

つけて消去できるような局所的な決定力をもった属性

および分割の基準点を，その都度ランダムに選択する．

注目履歴の先頭要素は，反復消去の最終期手前まで残

った候補たちであり，もし単一の候補であればこれを

最終選択とし，複数であればその中からランダム選択

する．付録Aおよび付録Bに反復消去アルゴリズムの

プログラムを掲載した（開発言語：SWI Prolog for 

Windows 64 bit version 8.2.1）．これらを実行することに

よって確率的注目を伴う反復消去の下での選択比率の

予測値が実験的に得られる．ただし便宜的に反復の深

さを 15回に制限した．Prologのバックトラック機能に

よって反復消去を自然にシミュレーションすることが

できる．また選択される代替案の比率は，すべてのバ

ックトラック分岐点において等確率を仮定したときの

理論分布と一致する．これは確率論理プログラミング

ないし点過程の一種として扱いうるが，本研究の分析

結果に直接影響しないためこれ以上触れない．  

ただし提案するアルゴリズムでは代替案間の類似性

に依存したサンプリングの調整を行う．似ている代替

案同士は似ていない代替案よりも頻繁に比較されると

いう経験的なルールである．属性ベクトル間のノルム𝑑

の一次変換に反比例して，付録 A の最後方にある

distribution_of_comparison述語のバックトラック回数を

調整する．その頻度は具体的には次の式(1)にしたがう． 

 ⌊𝛼 (𝛽 + 𝛾𝑑)⁄ ⌋ （1） 

ここで 3 つのパラメータは，標準的にα=2,β=0,γ=1

とする．なお式（1）は学習心理学や行動経済学におい

てマッチング法則として知られる． 

以降の部分では，4 領域の行動選択問題に順次提案

するモデルを適用する．第 2 節では推論課題の一種で

あるWasonのカード選択課題におけるマッチングバイ

アスを再現する．第 3 節では消費者ブランド選択にお

ける文脈効果，いわゆるビッグスリー（魅力効果，妥

協効果，類似効果）に対して現象の再現にとどまらず，

実証的な選択比率を再現する．両節を通じて反復消去

の挙動が具体的に理解されるはずである．第 4 節では

とくにアノマリーの種類が多く知られるギャンブル比

較の領域でのサンプル例題を集中的に実験する．また

時間選択問題についてもとりあげる．また第 5 節では

関連研究にふれる．最後に第 6節で結論を述べる． 

2. 認知モデルとしての反復消去の解釈 

本節ではWasonのカード選択課題として知られる例

題をとりあげ，回答者の認知プロセスに対応して，図

１の反復消去アルゴリズムがどのように適用され，そ

のパフォーマンスを再現するのか説明してみよう． 

Wason のカード選択課題[17]は次のような論理的推

論の問題である．机の上に 4 枚のカードが並べられ，

それぞれのカードの各面にはアルファベットの文字と

数字が記入されている．たとえば見えている文字が「Ｅ」

「Ｋ」「４」「７」であるとする．実験参加者は，カー

ドをめくって「もしカードの片面が母音ならば，もう

一方の面は偶数でなければならない」というルールが

正しいどうかを確かめる．ただし，調べる必要のない

カードを裏返さないよう指示される．論理的には，こ

のルールは前提 A（母音）と結論 C（偶数）とを結ぶ

条件命題 A→C，つまり「もし Aならば Cである」と

して理解され，その反例「Aでありかつ Cでない」を

検出するためには，A のカードであるカード「Ｅ」お

よびCの否定であるカード「７」を調べる必要がある．

ところが，「７」を調べる回答者は極めて少なく，その

正答率は 1 割程度とされる．誤答パタンの多くは母音

の「Ｅ」のみか，偶数の「４」との組み合わせである．

前提に一致するカードを選ぶ傾向は確証バイアスと呼

ばれた．しかしルールが抽象的であっても，「もし A

ならば Cではない」のように結論が否定表現されると

正答率は大きく改善されることから，抽象的なルール

では前提と結論に直接対応するカードが選ばれやすい

と考えられる．これはマッチングバイアスと呼ばれる．

ちなみにマッチングバイアスは，意思決定心理学にお

いて選好逆転現象などのアノマリーを説明するために

用いられた（尺度）両立可能性と呼ばれる仮説に対応

している． 

反復消去を適用すると，確証バイアスあるいはマッ

チングバイアスが再現される．アルゴリズムを適用す

る前に，カード選択課題の認知プロセスを計算的なメ

カニズムとして解釈しておこう．回答者は１（調べる）

または０（調べない）を割り当てる．全部で２の 4乗，

８通りの可能な回答パタンがあり，回答者はそのうち

の一つを選ぶことで正解（利得１）を得る．この問題

を解くため，回答者の心の中に，適切なメッセージプ

ロセス，あるいはメカニズムが構成される．すなわち，

まず問題文からルールを理解し，カード表面に書かれ

た文字を判読し，ルールに対してその違反を調べる必

要性を評価するための情報を得る．そしてこの情報に

反応して回答パタンが決定される． 

ここで Prologプログラムの使い方も含めて，反復消

去がどのようにコンピュータで自動実行されるのかに

ついて簡単に説明しておこう．付録の Prologプログラ

ムではカード選択問題における選択対象を次の 5 つの

事実節として表現する．  

data( wst( 2 ), [ 0, 0, 1, 1 ], 'Ｋ' ). 

data( wst( 2 ), [ 1, 10, 0, 0 ], 'Ｅ' ). 

data( wst( 2 ), [ 1 ,0, 1, 0 ], '６' ). 

data( wst( 2 ), [ 0, 10, 0, 1 ], '７' ). 

data( wst( 2 ), [ 0, 0, 0, 0 ], '－' ). 

上の各行，data( wst( 2 ), [ S, T, U, V ], X )における S～

V の４変数は属性ベクトル，各次元は（母音，偶数），

（母音，奇数），（子音，偶数），（子音，奇数）という

４種類のカードの可能な状態に対応する．属性値はカ

ードを調べたときの期待利得と解釈し，また変数Xは

回答者の選択に対応する．最後の'－'はいずれのカード

も選択しないケースである．各行はデータの単位とし

ての事実節として Prologシステムに登録される．  

ユーザーがプロンプト?- の後に目標節を入力する

ことはクエリー（質問）と呼ばれる．Prolog システム

は目標節を証明できる事実を見つけ出す． 

?- P=wason_selection_task(2), simulate_choices( P, [W], 

S, K, N, V ). 

P = wason_selection_task(2), 

W = S, S = '６', 

K = 3552, 

N = 20496, 

V = 0.17330210772833723 

セミコロン;を入力してバックトラックさせる． 

P = wason_selection_task(2), 

W = S, S = '７', 

K = 2600, 

N = 20496, 

V = 0.12685402029664325 ; 

P = wason_selection_task(2), 

W = S, S = 'Ｅ', 

K = 14344, 

N = 20496, 

V = 0.6998438719750195. 

変数 W（＝S），K，N，V の意味はそれぞれ，カー

ドのオモテ面，選択頻度，試行回数，選択率である．

選択率は「Ｅ」が 70%，「６」が 17%，「７」が 13%，

「Ｋ」が 0%である．  

同じ例題についての，より率直な問題表現としては，

属性値ベクトルをルールの前提と結論の成否に直接対

応させる． 

data( wst( 1 ), [ 0, 0 ], 'Ｋ' ). 

data( wst( 1 ), [ 1, 1 ], 'Ｅ' ). 

data( wst( 1 ), [ 1 ,0 ], '６' ). 

data( wst( 1 ), [ 0, 0 ], '７' ). 

data( wst( 1 ), [ 0, 0 ], '－' ). 

以下のクエリーを実行すると，クエリー末尾の fail

でバックトラックが強制されて，ｋ番目の属性 attr(k)

に注目したときの各代替案（カード）ペア間の優劣比

較がすべて書き出される．  

?- P= wason_selection_task(1), compare_attribute( P, A, E, 

C ), nl, write( wst(1);A;E;C), fail. 

 

wst(1);attr(1);eval(1,0);item(Ｅ>Ｋ) 

wst(1);attr(1);eval(1,0);item(Ｅ>６) 

wst(1);attr(1);eval(1,0);item(Ｅ>７) 

wst(1);attr(1);eval(1,0);item(Ｅ> －) 

wst(1);attr(2);eval(1,0);item(６>Ｋ) 

wst(1);attr(2);eval(1,0);item(６>Ｅ) 

wst(1);attr(2);eval(1,0);item(６>７) 

wst(1);attr(2);eval(1,0);item(６> －) 

false. 

以下のクエリーを実行すると，反復消去の系列が一

つ生成される． 

?- P = wason_selection_task(1), [ _ : [ S ] | _ ] = T, 

elimination_wrt_attribute( P, T ). 

P = wason_selection_task(1), 

S = 'Ｋ', 

T = [attr(1):['Ｋ'], attr(2):['Ｅ'], attr(1):['６'], attr(1):['７'], 

attr(1):[－]] . 

候補リスト T には，反復消去のステップごとに注目

された属性とそれによって消去された候補集合が，リ

ストの左端に順次追加されていく．上の実行例では，

３ステップ目までは 1 番目の属性が注目されて「Ｅ」

と「Ｋ」が候補に残るが，４ステップ目で 2 番目の属

性が注目されると「Ｅ」が消去されて，「Ｋ」が最終的

に選択される．すべての注目経路をバックトラックさ

せている．付録Aの attr_trace述語を用いると全注目経

路を収集できる．バックトラックさせると，最終選択

の比率は「Ｅ」と「６」が 29%，「K」と「７」および

無選択が 14%ずつとなる． 

3. 文脈効果 

マーケターは消費者に対して売れるブランド品（タ

ーゲット）を提供する役割を担う．マーケターは，魅

力効果を含めたビッグスリーと呼ばれる典型的な３タ

イプの文脈効果を熟知する必要がある．これらは消費

者ブランド選択における伝統的な理論（Luce モデル）

と矛盾することから，計量マーケティングや意思決定

心理学においてその代替理論が提案された． 

消費者ブランド選択は，より一般的に多属性意思決

定問題の一種であり，代替案（ブランドアイテム）は

複数の次元からなる属性値の組として評価されると仮

定され，評価の高いものがより選択されやすいものと

予測される．ビッグスリーと呼ばれる現象は，実験的

に想定される以下のような状況である．ターゲットと

拮抗する市場シェアを持つ競合品に対し，おとり商品

（デコイ）を２次元の属性空間のある位置に投入する

と，ターゲットの相対的魅力が増す．１）おとりを投

入するターゲットから見て劣った位置に追加する，２）

ターゲットを中間位置になるように両製品を結ぶ延長

線上におとりを追加する．３）おとりを競合品の近く

に追加してそのシェアを奪わせる．それぞれ１は魅力

効果，２は妥協効果（あるいは折衷効果），３は類似効

果と呼ばれる．  

魅力効果の例における属性への注目と反復消去の系

列を具体的に説明しよう．属性空間は値引の程度と品

質の高さという 2 次元の値の組として表現され，消費

者はこの情報に反応して選択を行うと仮定する．例と

して，例えば X：値引率，Y：品質とし，また３ブラ

ンドの評価値ペア（X，Y）がそれぞれA：（２割値引，

中品質），B：（定価，極上），C：（1 値割引，高品質）

とする．マーケッターは Bをターゲットとして競合品

Aよりも多く売りたい．消費者の参照点ａ0が「０割引」

とすると，時点ｔ＝１で「値引」が注目されたとき，

おとり商品Cは消去され，AとBが残る．続いてｔ＝

２でも再び「値引」が注目されたとするとＡが最後に

残る．しかしｔ＝２で「品質」が注目され，参照点が

「高品質」とすると最後に Bが残る．一方もしｔ＝１

で「品質」が注目されると，残るのは B と C になり，

ｔ＝２で「品質」あるいは「値引」のいずれが注目さ

れても，Bが選ばれる（図 2も参照）．ちなみにＢだけ

には劣った選択肢Ｃがある（つまり CはBに対して局

所的に非許容的である）とき，B がより選ばれやすく

なる．  

 

 

図 2  魅力効果を再現する属性注列とその確率 

 

図 2は属性の注目順序（縦棒の右側）の下で残った

代替案集合（縦棒の左側）を表す．注目される属性が

等確率で選ばれると仮定すると，Aの選択確率は 25%，

Bは 75%，Cは 0%となり，魅力効果が再現される（図

2参照）．類似効果は妥協効果と属性評価順序が同じで

あるため，順序だけでは再現できない．そこで代替案

の類似性を反映してモデルを修正する．付録Dにこの

アイディアを実行する Prologプログラムを示した．反

復消去はこのアイディアを精密化し，かつ文脈効果以

外にも適用できるように一般化したものである．  

 

表 1 文脈効果のサンプルデータ 

 ターゲット 競合品 おとり 

魅力効果 4, 8 8, 4 2, 6 

妥協効果 7, 4 5, 6 3, 8 

類似効果 8, 3 3, 8 4, 7 

 

表 1 に文脈効果の数値例を示す．また表 2 として付

録 A と付録 B に掲載した実験用プログラムを用いて，

文脈効果を再現した実験結果をまとめておく． 

 

表 2 シミュレーション実験の結果† 

 ターゲット 競合品 おとり 

魅力効果 68.2% 31.8% 0% 

妥協効果 36.4% 31.8% 31.8% 

類似効果 39.1% 26.1% 34.8% 

†
式(1)のパラメータは𝛼 = 2, 𝛽 = 0, 𝛾 = 1とし，prologのバックトラックによっ

て系統的サンプリングを行った． 

{A,B,C}
（100%）

{A,B}|X 
（50%）

{A}|XX

（25%）

{B}|XY

（25%）

{B,C}|Y 
（50%）

{B}|YX 

（25%）

{B}|YY

（25%）

図 3  ビッグスリーのシミュレーション実験 

 

図 3はWeb上の Prolog処理系である SWISH Prolog

上で実際に文脈効果のシミュレーション実験を行った

画面である 

最後に，両極の代替案Aと Bを固定し，Cの位置を

変化させたときに優勢になる代替案のフェイズダイヤ

グラムを図4に示す（文献[2]の図 13と比較されたい）． 

 

 

図 4  おとりの追加位置と文脈効果 

4. ギャンブル比較，時間選択 

本節では 4 領域からギャンブル比較に焦点を絞り，

サンプル例題を集中的に採取する．ギャンブル比較で

は代替案はすべて可能な賞金金額とその確率を組み合

わせたものである．実験では問題ごとに金額，確率，

金額と確率の乗算（期待値），参照点およびこれらの順

序といった属性へのマッピングを工夫した．またいず

れも二択問題であるので，参照点として非選択に相当

する第 3の代替案を設定した（詳細は付録 Cを参照さ

れたい）．  

実験に用いるサンプル例題は以下の通りである． 

イ) common_ratio_effect：アレの背理[1] 

ロ) common_consequence_effect：アレの背理[1] 

ハ) reflection_effect：鏡映効果[7] 

ニ) possibility_effect：可能性効果[7] 

ホ) isolation_effect：孤立効果[7] 

ヘ) ellsberg_paradox：エルズバーグの背理[4] 

ト) branch_splitting_effect：事象分岐効果[3] 

チ) preference_reversals: 選好逆転現象[8] 

リ) framing_effect: フレーミング効果[15] 

ヌ) time_preference: 現在バイアス[11] 

最後の例題を除くといずれも人々の典型的な選択パ

タンが期待効用最大化の仮定に矛盾する．これらは従

来それぞれ機序の異なるアノマリーとして論じられて

きた．たとえばアレの背理（確実性効果と共通比によ

る確率縮小あるいは共通結果削除による確実性効果の

消失），鏡映効果，可能性効果などは，独立性公理への

違反だが，エルズバーグの背理は確率のあいまい性を

嫌う傾向，孤立効果は二段階クジにおける疑似的な確

実性効果である．現在バイアスは，ギャンブル比較で

はなく，報酬に時間遅れがある選択問題におけるアノ

マリーであり，指数割引と矛盾する（双曲割引を示す）

が，その逸脱パタンはギャンブル比較のアノマリーの

それと対応する[11]．表 3に反復消去アルゴリズムを用

いた実験結果をまとめた． 

 

表 3  ギャンブル比較のアノマリーの実験結果 

 

表 3 の各行はそれぞれの例題のモデル識別名であり，

各列に 3 代替案の相対的選択頻度（分数）とそのパー

セントのデータが示される．空白部分はゼロ件であり，

その代替案の選択データは生成されなかったことを示

す．カッコ中の数値が 1 と 2，または 3 と 4 の行ペア

で代替案 1 と代替案 2 の間の選択率の大小が逆転する

パタンがアノマリー，つまり理論モデルへの違反であ

る．ニ以外で識別名のカッコ内が 3 や 4 である行は，

属性マッピングを変更した同じ例題のモデルである．

また実験では文献を参考に再現目標となる選択率を設

定した．典型的とされた回答については表 3 の乖離の

列に実験で得られた相対選択率δから目標値γを引い

た差δ－γを示した．表 3 末尾の列λは，選択率 50%

を帰無仮説としてベルヌーイ試行 100 回のときの片側

累積確率の比，P（δ＋非選択率×0.5）/ P（γ）であ

る．λの大きさはモデルの適合の直観的な目安となる．

λ＝１の近くは実証値がよく近似され，λ＜１のとき

はより帰無仮説に近く，λ＞１のときはバイアスをよ

り強く予測する．λ＜0 のときは選択傾向の予測に失

敗している．表 3からわかるように，ニ（可能性効果）

とヌ（現在バイアス）の 2 と 4，へ（エルズバーグ背

理）の 1 と 3 でλが負であるが，選択変化の傾向は再

現されている．またヌでは多数派は時間整合という実

証的事実と符合する．それ以外はおおむね良い．ニに

ついては確率が小さく反復回数が突出しており，類似

度のカットオフ閾を 0から 0.01に変更することによっ

て近似が改善する（λ =0.91, 1.03, 1.00, 1.00）． 

?- stat_choice( possibility_effect( _ ) ). 

fit: 1.000 [4/4] 

possibility_effect(1);Ａ;40/96 [0.417] 

possibility_effect(1);Ｂ;56/96 [0.583] Dif:-0.217 

possibility_effect(2);Ａ;72/112 [0.643] Dif:0.043 

possibility_effect(2);Ｂ;40/112 [0.357] 

possibility_effect(3);Ａ;40/112 [0.357] 

possibility_effect(3);Ｂ;72/112 [0.643] Dif:-0.217 

possibility_effect(4);Ａ;72/112 [0.643] Dif:-0.087 

possibility_effect(4);Ｂ;40/112 [0.357] 

true. 

類似度のカットオフは，いわば排斥項付の結合エネ

ルギー（LJポテンシャル）の簡便な代用である． 

5. 関連研究 

確率的選択モデルは，意思決定論ではランダム効用

とも呼ばれ，Luce によって公理化された．McFadden

によってマーケティング分野や交通工学へ応用されロ

ジットモデルとして計量分析者の間でおなじみである．

学習心理学において数学的に同等のモデルは古くは

Thurstone によって，後に Herrnstein らによってマッチ

ング法則として論じられた．後者は時間割引のアノマ

リーである現在バイアスを再現する（双曲型割引）．そ

れゆえ魅力効果はマッチング法則の違反でもある．こ

れはブランドから感じる魅力は単純に代替案の評価値

に対する反応ではないことを示唆する．  

確率的な属性への注目を通じて代替案を消去する多

属性意思決定モデルとしては，Tverskyらによって提案

された属性による消去（elimination by aspects; EBA）が

知られている[14]．EBA は文脈効果の研究の先駆けと

なった． EBAでは情報反応（属性へのランダム注目）

にかんしてはマッチング法則（Luceの公理）が成り立

つ．本論文の提案するモデルは属性に確率的に注目し

て反復消去を行うことと，類似性に依存して比較頻度

が修正されるという点では，EBAと発想を同じくする．

しかし EBA では類似効果を再現できるが魅力効果や

妥協効果を再現できない．反復消去はビッグスリーを

その実証比率まで正確に予測することができる． 

Tversky, Sattath, and Slovic[16]は反復消去に似たヒュ

ーリスティック手続きを考察しており，またその心理

学的要因を検証している．２案の評がバランスするよ

う未定の属性値を埋める手続き（マッチング）と比べ

て，代替案を選択する手続きでは優先的に注目された

属性が決定を左右する傾向が強まる．その結果として

手続き不変性に違反する．これはプロミネンス仮説と

呼ばれ，両立可能性の特殊ケースである．  

反復消去アルゴリズムは，特定の属性に注目する傾

向を持ったエージェントの集合体を考え，エージェン

トのランダム選択をメッセージとし，選択対象の集合

を絞り込むような情報反応あるいはそれを引き起こす

認知的なメカニズムとして解釈されうる．意識的な情

報処理が省かれるという観点では，限定された合理性

（bounded rationality）のモデルを提供する．分散化され

た認識行動サイクルでは，エージェントの行動が，他

のエージェントを含む環境の残りの部分へのメッセー

ジとなり，情報反応の連鎖が引き起こされる．このよ

うな分権化（decentralization）されたメカニズムは，ゲ

ーム理論やマルチエージェントシステムの AI 研究で

よく研究されており，意思決定の認知モデル化にも応

用されている． 

ちなみに投票理論[13]において反復消去手続きは，ヘ

ア(Hare)方式と呼ばれる．情報反応モデルは，その単純

なバージョンとして，意思決定者の認知プロセスを単

一のエージェントからなるメカニズムの下で説明する

が，これをエージェントの集団による組織的メカニズ

ム，あるいは「心の社会」へと自然に拡張することが

できる．ちなみにVAMPモデル[2]ではビッグスリーの

再現のため Saariの投票方式を用いる．すでに述べたよ

うに，類似性を考慮したランダム生成を，推移性を保

つようにするだけでビッグスリーは再現される．つま

りCondorcet方式でも同様に可能である．安定結婚問題

を解く遅延受入れ方式（deferred acceptance），いわゆる

DA アルゴリズム[5]はブロッキングペアを見つけて反

復消去する．安定結婚問題における「男性」を「注目

属性」，「女性」を「消去候補」と読み替えることで，

反復消去の分権化されたバージョンが得られる． 

6. 結論 

本研究では属性への確率的な注目と代替案の反復的

消去を用いて意思決定者の認知プロセスのモデル化を

提案し，４領域計 14のアノマリーに対して実証的な回

答傾向を予測できるか検証した． 

反復消去アルゴリズムでは注目属性と消去候補がラ

ンダム選択される．類似性によるバックトラック頻度

調整（(1)式）以外に心理学的なバイアスが入り込む余

地はない．このことは次のような率直な疑問を投げか

ける．心理学者や行動経済学者らがバイアスと呼んで

いるものの多くは，本質的にランダムなノイズと区別

できないのではないか．反復消去アルゴリズムは，こ

れらの選択問題の領域ごとに考案されたあらゆる特殊

モデルに対し，いわば洗練された帰無仮説を提起する

という意味において認知モデルとしての汎用性をもつ．

またそこからもう一つの疑問が呈される．すなわち認

知プロセスの基本的なアルゴリズムは領域固有的に進

化した集合体ではなく，単一的であり，問題タイプや

行為の状況に応じて問題の表象の仕方が管理されてい

るのではないか． 

本研究で提案されたモデルはあくまで萌芽的であり，

予測の当てはまりは限定的である．消去において切断

する集合に再注目を許すと若干の改善が得られるが，

計算効率が悪い．また限られた少数の例題に適用した

にとどまる．例えばモンティホール問題のような確率

判断の領域のアノマリーは本研究で論じられなかった．

これについては読者の練習とされたい．またこれは属

性値ベクトルのマッピングがアドホックであることと

関連する．属性マッピングを系統化してより多くの例

題に適用したとき，精度だけでなく，多様な表象方法

の下でも予測が大きくぶれないことが期待される．こ

の観点での表象安定性は，規範的な意思決定モデルの

手続き不変性と対比されよう．また，提案されたモデ

ルでは同一意思決定者内での一貫性と他の課題への転

移可能性を論じていない．バックトラックの制限

（Prologのカット演算のようなもの）を個人間に分散

させる解釈は可能だが，被験者間アンサンブルが等確

率の事前分布に一致することを説明する必要が生じる．

こうした課題を残しつつ，反復消去による認知的メカ

ニズムを探求する意義はあろうかと思われる． 

付録 A．反復消去のPrologプログラム 

 

付録 B．実験用スクリプト 

 

choice_problem( context( A ), "文脈効果のビッグスリー" ):- 
  A = attraction_effect; 
  A = compromise_effect; 
  A = similarity_effect. 

item_list( context( A ), [ 'Ａ', 'Ｂ', 'Ｃ' ] ):- choice_problem( context( A ), _ ). 
choice_item( P, X ):- 

  choice_problem( P, _ ),  
  item_list( P, A ),  
  member( X, A ). 

brand(attraction_effect,[8,4],'Ａ'). 
brand(attraction_effect,[4,8],'Ｂ'). 
brand(attraction_effect,[2,6],'Ｃ'). 
brand(compromise_effect,[7,4],'Ａ'). 
brand(compromise_effect,[5,6],'Ｂ'). 
brand(compromise_effect,[3,8],'Ｃ'). 
brand(similarity_effect,[8,3],'Ａ'). 
brand(similarity_effect,[3,8],'Ｂ'). 
brand(similarity_effect,[4,7],'Ｃ'). 
branch( context( A ), Pos, Item ):-  brand( A, Pos, Item ). 
branch( Issue, Pos, Item ):-  data( Issue, Pos, Item ). 
compare_attribute( Problem, attr(J), eval(P, Q), item(K > M) ):-  

  branch( Problem, X, K ), 
  branch( Problem, Y, M ), 
  nth1( J, X, P ), 
  nth1( J, Y, Q ), 
  0 < P - Q. 

attribute( attr(1) ). 
attribute( attr(2) ). 
merit_on_attribute( P, Attr, X, Y, I, J ):-  

  attribute( Attr ), 
  member( I, X ), 
  member( J, Y ), 
  compare_attribute( P, Attr, _, item( I > J ) ). 

limit_tree_length( 15 ). 
constraint_on_limit_tree_length( B ):- 

  limit_tree_length( L ), 
  length( B, K ), 
  K < L. 

exclusive_in_groups_if_gamble_branches( X, _ ):- 
  ¥+ member( _: _: _, X ), 
  !. 

exclusive_in_groups_if_gamble_branches( X, Y ):- 
  findall( J, member( J: _: _, X ), A ), 
  findall( J, member( J: _: _, Y ), B ), 
  intersection( A, B, C ), 
  ( K =1, C = [ ] -> between( 1, K, _ ) ; true ). 

list_projection( [ ], [ ], [ ]). 
list_projection( [ 0 | A ], [ _ | Y ], Z ):- list_projection( A, Y, Z ). 
list_projection( [ 1 | A ], [ X | Y ], [ X | Z ] ):- list_projection( A, Y, Z ). 
take_or_leave_function( A, _, X, _, Y ):- 

  list_projection( _, A, X ), subtract( A, X, Y ). 
divided_into_two_wrt_attr( P, H, Attr, X, Y ):- 

  take_or_leave_function( H, _, X, _, Y ),  
  X ¥= [ ], 
  exclusive_in_groups_if_gamble_branches( X, Y ), 
  attribute( Attr ), 
  ¥+ merit_on_attribute( P, Attr, X, Y, _, _ ). 

elimination_wrt_attribute( _, [ ], [ H | B ], [ H | B ] ):- H = _ : [ _ ]. 
elimination_wrt_attribute( P, A, B, C ):- 

  constraint_on_limit_tree_length( B ), 
  divided_into_two_wrt_attr( P, A, D, X, Y ), 
  distribution_of_comparison( P, D, X, _ ),  
  elimination_wrt_attribute( P, Y, [ D : X | B ], C ). 

elimination_wrt_attribute( P, T ):- 
  choice_problem( P, _ ), 
  findall( X, ( 
   choice_item( P, X ), ¥+ ¥+ branch( P, _, X ) 
  ), A ), 
  elimination_wrt_attribute( P, A, [ ], T ). 

distances_in_item_list( P, D, X, L ):- 
  choice_problem( P, _ ), 
  attribute( D ), 
  findall( E, ( 
   compare_attribute( P, D, eval( U, V ), item( A > B ) ), 
   subset( [ A, B ], X ), 
   E is U - V 
  ), L ). 

average_distance_in_item_list( P, D, X, V ):- 
  distances_in_item_list( P, D, X, U ), 
  sum_list( U, T ), 
  length( X, K ), 
  V is T / K. 

:- dynamic model_parameters/1. 
model_parameters( [ 2, 0, 1 ] ). 
repetition( N ):- between( 1, N, _ ). 
distribution_of_comparison( P, D, X, _H ):- 

  average_distance_in_item_list( P, D, X, V ), 
  model_parameters( [ Prm1, Prm2, Prm3 ] ), 
  Q0 is ( Prm2 + Prm3 * V ), 
  %( ( V = 0.0; Q0 = 0 )  -> K = 0  
  ( ( V < 0.01; Q0 = 0 )  -> K = 0  
  ; K is integer( Prm1 / Q0 ) 
  ),  F is max( 1, K ), 
  repetition( F ), !. 

distribution_of_comparison( _, _, _, 1 ). 
 

%各代替案が選出されるすべての注目経路を調べて集計する． 
attr_trace(P):- 

 findall( S;T, ( 
elimination_wrt_attribute( P, T ), 
[ _ : [ S ] | _ ] = T 

 ), L ), 
 bagof( 1, member( S;T, L ), F ), 
 length( F, N ), 
 tab(2), 
writeln( S; T; N ), 

 fail; true. 
 
/* デモ実行： 
?- P = context(attraction_effect), attr_trace(P). 
Ａ;[attr(1):[Ａ],attr(1):[Ｂ],attr(1):[Ｃ]];8 
Ａ;[attr(1):[Ａ],attr(1):[Ｂ,Ｃ]];6 
Ａ;[attr(2):[Ａ],attr(1):[Ｂ],attr(1):[Ｃ]];8 
Ａ;[attr(2):[Ａ],attr(1):[Ｂ,Ｃ]];6 
Ｂ;[attr(1):[Ｂ],attr(1):[Ｃ],attr(2):[Ａ]];8 
Ｂ;[attr(1):[Ｂ],attr(2):[Ａ],attr(1):[Ｃ]];8 
Ｂ;[attr(1):[Ｂ],attr(2):[Ａ,Ｃ]];6 
Ｂ;[attr(1):[Ｂ],attr(2):[Ｃ],attr(2):[Ａ]];8 
Ｂ;[attr(2):[Ｂ],attr(1):[Ｃ],attr(2):[Ａ]];8 
Ｂ;[attr(2):[Ｂ],attr(2):[Ａ],attr(1):[Ｃ]];8 
Ｂ;[attr(2):[Ｂ],attr(2):[Ａ,Ｃ]];6 
Ｂ;[attr(2):[Ｂ],attr(2):[Ｃ],attr(2):[Ａ]];8 

P = context(attraction_effect). 
?- 
*/ 

simulate_choices( P, W, S, K, N, V ):- 
  choice_problem( P, _ ), 
  findall( P; S, (  
   elimination_wrt_attribute( P, T ),  
   [ _ : [ S ] | _ ] = T  
  ), L ),   
  length( L, N ),  
  member( S, W ), 
  bagof( 1, member( P; S, L ), U ),  
  length( U, K ),  
  V is K / N.  

rating_fitness( L, M, K, N ):- 
  findall( 1, ( member( [ Y | _ ], L ), Y == 1 ), A ), 
  length( A, M ), 
  findall( 1, ( member( [ Y | _ ], L ), Y == 0 ), B ), 
  length( B, K ),  
  N is M + K. 

rating_fitness( L, M / N = F ):- 
  rating_fitness( L, M, K, N ), 
  N > 0, !, F is M / ( M + K ), 
  format( 'fit: ~3f', F ), tab( 1 ), write( [ M / N ] ). 

rating_fitness( _, _ ). 
evaluate_prediction( thres( Thres ), V, D, Y ):- 

  D is V - Thres,  
  ( D < 0 -> Y = 0 ; Y = 1 ).  

evaluate_prediction( P, 'Ｂ', V, D, Y ):- 
  P = context( attraction_effect ), 
  evaluate_prediction( thres( 0.6 ), V, D, Y ).  

evaluate_prediction( P, 'Ｂ', V, D, Y ):- 
  P = context( compromise_effect ), 
  evaluate_prediction( thres( 0.35 ), V, D, Y ).  

evaluate_prediction( P, 'Ａ', V, D, Y ):- 
  P = context( similarity_effect ), 
  evaluate_prediction( thres( 0.35 ), V, D, Y ).  

evaluate_prediction( P, X, V, D, Y ):- 
  P ¥= context( _ ), 
  evaluate_prediction_x( P, X, V, D, Y ). 
evaluate_prediction( _, _, _, _, _ ). 
evaluate_predictions( Z, W, L ):- 

 findall( [ Y, S, P, K, N, V, D ], ( 
  member( P, Z ), 
  simulate_choices( P, W, S, K, N, V ), 
  once( evaluate_prediction( P, S, V, D, Y ) ) 
 ), L ). 

write_error_term( S, D, Y ):- 
  ¥+ ( S = option( K ), K ¥= 2 ), 
  ¥+ var( Y ), format( ' Dif:~3f', D ), Y == 0, write( * ), !. 

write_error_term( _, _, _ ). 
write_prediction_fitness_for_each_branch( L ):- 

  member( [ Y, S, P, K, N, V, D ], L ), 
  nl, write( P; S; K/N ), tab(1),  format( '[~3f]', V ),  
  write_error_term( S, D, Y ). 

stat_choices( Z, W ):- 
  evaluate_predictions( Z, W, L ), 
  rating_fitness( L, _ ), 
  write_prediction_fitness_for_each_branch( L ),  
  fail. 

stat_choices( _, _ ). 
stat_choice( A ):- stat_choices( [ A ], [ _ ] ). 

付録C．選択問題の属性表現データ  

付録D．文脈効果を再現する代替的方法 

 

 

選択('A',P):- 順位('AC',P), 順位('CB',P), between( 1, 2, _ ). 
選択('A',P):- 順位('AC',P), 順位('BC',P). 
選択('B',P):- 順位('BC',P), 順位('CA',P), between( 1, 2, _ ). 
選択('B',P):- 順位('AC',P), 順位('BC',P). 
選択('C',P):- 順位('CB',P), 順位('CA',P), between( 1, 2, _ ). 
サンプル発生( _百回中の度数, Id ):- between(1, _百回中の度数, Id ). 
順位('AC',魅力効果) :- サンプル発生( 100, _ ). 
順位('CA',魅力効果) :- サンプル発生( 0, _ ). 
順位('BC',魅力効果) :- サンプル発生( 50, _ ). 
順位('CB',魅力効果) :- サンプル発生( 50, _ ). 
順位('BC',妥協効果):- サンプル発生( 50, _ ). 
順位('CB',妥協効果):- サンプル発生( 50, _ ). 
順位('CA',妥協効果):- サンプル発生( 45, _ ). 
順位('AC',妥協効果):- サンプル発生( 55, _ ). 
順位('BC',類似効果):- サンプル発生( 50, _ ). 
順位('CB',類似効果):- サンプル発生( 50, _ ). 
順位('CA',類似効果):- サンプル発生( 55, _ ). 
順位('AC',類似効果):- サンプル発生( 45, _ ). 
問題タイプ( 魅力効果, ターゲット( 'A' ) ). 
問題タイプ( 妥協効果, ターゲット( 'A' ) ). 
問題タイプ( 類似効果, ターゲット( 'B' ) ). 
実験( P, T, [ A, B, C ], N, [ RA, RB, RC ] ):- 
  問題タイプ( P, T ), 
  findall( 1, 選択('A',P), L1 ), 
  findall( 1, 選択('B',P), L2 ), 
  findall( 1, 選択('C',P), L3 ), 
  length( L1, A ), 
  length( L2, B ), 
  length( L3, C ), 
  N is A + B + C, 
  RA is A / N,  
  RB is B / N, 
  RC is C / N. 

 
/* デモ実行 
?- 実験( P, T, D, N, R ), nl, write( P; T; N; D; R ), fail. 
魅力効果;ターゲット(A);20000;[15000,5000,0];[0.75,0.25,0] 
妥協効果;ターゲット(A);20000;[8250,7250,4500];[0.4125,0.3625,0.225] 
類似効果;ターゲット(B);20000;[6750,7750,5500];[0.3375,0.3875,0.275] 
false 
*/ 

%data( common_ratio_effect,[q$z, $x,$y,p$x],_). 
data( common_ratio_effect( 1 ),[0, 400,  0, 80],'Ａ'). 
data( common_ratio_effect( 1 ),[20, 300,300,100],'Ｂ'). 
data( common_ratio_effect( 1 ),[0, 0,0,0],'Ｃ'). 
data( common_ratio_effect( 2 ),[75, 400,0,20],'Ａ'). 
data( common_ratio_effect( 2 ),[80, 300,0,25],'Ｂ'). 
data( common_ratio_effect( 2 ),[0, 0,0,0],'Ｃ'). 
%data( common_consequence_effect,[q$z,$x,$y,$z,p$x],_). 
data( common_consequence_effect( 1 ),[0, 500,100,  0, 10],'Ａ'). 
data( common_consequence_effect( 1 ),[1, 100,100,100,100],'Ｂ'). 
data( common_consequence_effect( 1 ),[0, 0,0,0,0],'Ｃ'). 
data( common_consequence_effect( 2 ),[89, 500,0,0,10],'Ａ'). 
data( common_consequence_effect( 2 ),[90, 100,0,0,11],'Ｂ'). 
data( common_consequence_effect( 2 ),[0, 0,0,0,0],'Ｃ'). 
%data( reflection_effect,[$x, -p$x, $y, p$y],_). 
data( reflection_effect( 1 ),[ -400, -80, 0, 20 ],'Ａ'). 
data( reflection_effect( 1 ),[ -300,-100, 0, 0 ],'Ｂ'). 
data( reflection_effect( 1 ),[ -600,-100, 0, 0 ],'Ｃ'). 
data( reflection_effect( 2 ),[ -400, -20, 0, 80],'Ａ'). 
data( reflection_effect( 2 ),[ -300, -25, 0, 75],'Ｂ'). 
data( reflection_effect( 2 ),[ -400, -100, 0, 0],'Ｃ'). 
%data( possibility_effect,[$x, p$x, $y, p$y],_). 
data( possibility_effect( 1 ),[400, 0.80,  0, -0.20],'Ａ'). 
data( possibility_effect( 1 ),[300, 1.00,  0, -0.00],'Ｂ'). 
data( possibility_effect( 1 ),[0, 0, 0, 0],'Ｃ'). 
data( possibility_effect( 2 ),[400, 0.008, 0, -0.992],'Ａ'). 
data( possibility_effect( 2 ),[300, 0.010, 0, -0.990],'Ｂ'). 
data( possibility_effect( 2 ),[300, 0, 0, -1],'Ｃ'). 
%data( possibility_effect,[$x, p$x, $y, p$y],_). 
data( possibility_effect( 3 ),[6000, 0.45,  0, -0.55],'Ａ'). 
data( possibility_effect( 3 ),[3000, 0.90, 0, -0.10],'Ｂ'). 
data( possibility_effect( 3 ),[0, 0, 0, 0],'Ｃ'). 
data( possibility_effect( 4 ),[6000, 0.001,  0, -0.999],'Ａ'). 
data( possibility_effect( 4 ),[3000, 0.002, 0, -0.998],'Ｂ'). 
data( possibility_effect( 4 ),[3000, 0, 0, -1],'Ｃ'). 
%data( isolation_effect,[$x, $y1, p$y1, $y2, q$y2],_). 
data( isolation_effect(1), [1000, 1000, 0.50, 0, 0.00],'Ａ'). 
data( isolation_effect(1), [1000,  500, 1.00, 0, 0.50],'Ｂ'). 
data( isolation_effect(1), [0, 0, 0, 0, 0],'Ｃ'). 
%data( isolation_effect,[$x+y1, p$y1, $x+y2, q$y2],_). 
data( isolation_effect(2), [1000 +1000, 0.50, 1000+0, 0.00],'Ａ'). 
data( isolation_effect(2), [1000 +500, 1.00, 1000+0, 0.50],'Ｂ'). 
data( isolation_effect(2), [0, 0, 0, 0],'Ｃ'). 
%data( isolation_effect,[$x+y1, p$y1, $x+y2, q$y2],_). 
data( isolation_effect(3), [1000 +1000, 0.50, 1000+0, 0.00],'Ａ'). 
data( isolation_effect(3), [1000 +500, 1.00, 1000+0, 0.50],'Ｂ'). 
data( isolation_effect(3), [1000 +0, 0, 0, 0],'Ｃ'). 
%data( ellsberg_paradox,[Pmax$, Pmin$, $red, $yellow, $blue],_). 
data( ellsberg_paradox(1), [1/3, 1/3, 1000, 0, 0],'Ａ'). 
data( ellsberg_paradox(1), [2/3, 0, 0,  1000, 0],'Ｂ'). 
data( ellsberg_paradox(1), [0, 0, 0, 0, 0],'Ｃ'). 
data( ellsberg_paradox(2), [1, 1/3, 1000, 0, 1000],'Ａ'). 
data( ellsberg_paradox(2), [2/3, 2/3, 0,  1000, 1000],'Ｂ'). 
data( ellsberg_paradox(2), [0, 0, 0, 0, 0],'Ｃ'). 
%data( ellsberg_paradox,[P$1max/min, $1, Q$2max/min],_). 
data( ellsberg_paradox(3), [1/3, 1/3, 1000, 1/3, 1],'Ａ'). 
data( ellsberg_paradox(3), [2/3, 0, 1000, 2/3, 2/3],'Ｂ'). 
data( ellsberg_paradox(3), [0, 0, 0, 0, 0 ],'Ｃ'). 
data( ellsberg_paradox(4), [1/3, 1, 1000,  1/3, 1/3 ],'Ａ'). 
data( ellsberg_paradox(4), [2/3, 2/3, 1000, 1/3, 2/3],'Ｂ'). 
data( ellsberg_paradox(4), [0, 0, 0, 0, 0],'Ｃ'). 
%data( framing_effect,[結果1, Pr$結果1, 結果2, Pr$結果2],_). 
data( framing_effect( 1 ),[ 200, 1,  0,  0 ],'Ａ'). 
data( framing_effect( 1 ),[ 600, 1/3, 0, 2/3 ],'Ｂ'). 
data( framing_effect( 1 ),[ 0, 0, 0, 0 ],'Ｃ'). 
data( framing_effect( 2 ),[ -400, -1, 0, 0 ],'Ａ'). 
data( framing_effect( 2 ),[ 0, 1/3, -600, -2/3 ],'Ｂ'). 
data( framing_effect( 2 ),[ -400, -1, -600, -1 ],'Ｃ'). 
%data( branch_splitting_effect( 1 ),[$96,$90,$14,$12],_). 
data( branch_splitting_effect( 1 ),[90,0,5,5],'Ａ'). 
data( branch_splitting_effect( 1 ),[85,5,0,10],'Ｂ'). 
data( branch_splitting_effect( 1 ),[0,0,0,0],'Ｃ'). 
%data( branch_splitting_effect( 2 ),[$96,$96,$90,$14,$12,$12],_). 
data( branch_splitting_effect( 2 ),[85,5,0,5,5,0],'Ａ'). 
data( branch_splitting_effect( 2 ),[85,0,5,0,5,5],'Ｂ'). 
data( branch_splitting_effect( 2 ),[0,0,0,0,0,0],'Ｃ'). 
%data( branch_splitting_effect( 3 ),[$100,$50,$7],_). 
data( branch_splitting_effect( 3 ),[80,20,0],'Ａ'). 
data( branch_splitting_effect( 3 ),[95,0,5],'Ｂ'). 
data( branch_splitting_effect( 3 ),[0,0,0],'Ｃ'). 
%data( branch_splitting_effect( 4 ),[$100,$100,$50,$50,%7],'Ｇ1'). 
data( branch_splitting_effect( 4 ),[80,0,15,5,0],'Ａ'). 
data( branch_splitting_effect( 4 ),[80,15,0,0,5],'Ｂ'). 
%data( preference_reversals,[$x, $y, p$x],_). 
data( preference_reversals( 1 ),[4 * 99, -4 * 1],'Ａ'). 
data( preference_reversals( 1 ),[16 * 33, -2 * 67],'Ｂ'). 
data( preference_reversals( 1 ),[0, 0],'Ｃ'). 
data( preference_reversals( 2 ),[4, -4, 99, 67],'Ａ'). 
data( preference_reversals( 2 ),[16, -2, 33, 1],'Ｂ'). 
data( preference_reversals( 2 ),[0, 0, 0, 0],'Ｃ'). 
%つづく 

%つづき 
%data( time_preference,[delay$x, $x],_). 
data( time_preference( 1 ),[0, 1000],'Ａ'). 
data( time_preference( 1 ),[-7, 2000],'Ｂ'). 
data( time_preference( 1 ),[0, 0],'Ｃ'). 
data( time_preference( 2 ),[-365, 1000],'Ａ'). 
data( time_preference( 2 ),[-365-7, 2000],'Ｂ'). 
data( time_preference( 2 ),[0, 0],'Ｃ'). 
%data( time_preference,[reference, delay$x, $x],_). 
data( time_preference( 3 ),[0, 0, 1000],'Ａ'). 
data( time_preference( 3 ),[0, -7, 2000],'Ｂ'). 
data( time_preference( 3 ),[0, 0, 0],'Ｃ'). 
data( time_preference( 4 ),[-365, 0, 1000],'Ａ'). 
data( time_preference( 4 ),[-365, -7, 2000],'Ｂ'). 
data( time_preference( 4 ),[0, 0, 0],'Ｃ'). 
 
/ *デモ実行 
?- stat_choice( wason_selection_task( _ ) ). 
fit: 0.500 [2/4] 
wason_selection_task(1);－;8192/59280 [0.138] 
wason_selection_task(1);６;17352/59280 [0.293] Dif:0.014 
wason_selection_task(1);７;8192/59280 [0.138] 
wason_selection_task(1);Ｅ;17352/59280 [0.293] Dif:-0.196* 
wason_selection_task(1);Ｋ;8192/59280 [0.138] 
wason_selection_task(2);６;3552/20496 [0.173] Dif:-0.106* 
wason_selection_task(2);７;2600/20496 [0.127] 
wason_selection_task(2);Ｅ;14344/20496 [0.700] Dif:0.211 
true. 
 

注．６とＥの行尾のDif値は予め設定した目標値からの乖離である．この
例題ではEvans & Lynch(1973)の実験値より「Ｅ」は 21/43，「６」は 12/43
とした．*/ 

